# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Maintainer: Charles Mauch <cmauch@gmail.com>

pkgname=perl-html-tableextract
pkgver=2.15
pkgrel=6
pkgdesc="Perl/CPAN Module HTML::TableExtract : Flexible HTML table extraction"
arch=('any')
url="https://search.cpan.org/~msisk/HTML-TableExtract/"
license=("GPL" "PerlArtistic")
makedepends=("perl-html-parser" "perl-html-element-extended")
depends=("perl-html-parser" "perl-html-element-extended")
source=("https://www.cpan.org/authors/id/M/MS/MSISK/HTML-TableExtract-$pkgver.tar.gz")
sha256sums=('86c59c9d58da3ca174da5e62f5a0fb02f4da02b1b1e01df9e5d14bb65e4c3ecf')

build() {
  cd "$srcdir"/HTML-TableExtract-$pkgver
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}
package() {
  cd "$srcdir"/HTML-TableExtract-$pkgver
  make install DESTDIR="$pkgdir"
  find "$pkgdir" -name '.packlist' -delete
  find "$pkgdir" -name '*.pod' -delete
}
